
Fortran as a language

Short history of Fortran

John W. Backus et al (1954): The IBM Mathematical Formula Translating
System
Early years development: Fortran II (1958), Fortran IV (1961) Fortran 66 &
Basic Fortran (1966)
Fortran 77 (1978)
Fortran 90 (1991) a major revision and Fortran 95 (1997) a minor revision
to it

Short history of Fortran

Fortran 2003 (2004): a major revision, adding e.g. object-oriented
features, C-bindings

”Fortran 95/2003” is the current de facto standard
Fortran 2008: a minor revision, adding clarifications and corrections to
2003

Parallel computing using coarrays
The latest standard is Fortran 2018: a minor revision

Fortran coarray improvements
C interoperability improvements

Transition from source code to executable program

source code
(.f, .f90, .F, .F90)

modules
compiler output

(optional)

linker output
(optional)

executable

object code (.o, .so)

linker

compiler

libraries

Transition from source code to executable program

Simple programs can be compiled and linked in one go:
$ gfortran -o hello hello.f90 

$ ./hello 

Hello from Fortran

More complex cases will be discussed later on

Look and feel

program square_root_example

  ! comments start with an exclamation point.

  ! you will find data type declarations, couple arithmetic operations

  ! and an interface that will ask a value for these computations. 

  implicit none

  real :: x, y 

  intrinsic sqrt ! fortran standard provides many commonly used functions 

 

  ! command line interface. ask a number and read it in 

  write (*,*) 'give a value (number) for x:'

  read (*,*) x

  y = x**2+1 ! power function and addition arithmetic 

  write (*,*) 'given value for x:', x 

  write (*,*) 'computed value of x**2 + 1:', y 

  ! print the square root of the argument y to screen 

  write (*,*) 'computed value of sqrt(x**2 + 1):', sqrt(y)

end program square_root_example

Source code remarks

Free format source code, but
A variable name can be no longer than 31 characters containing only
letters, digits or underscore, and must start with a letter
Maximum row length is 132 characters

No distinction between lower and upper case characters
Character strings are case sensitive

Line break is the statement separator
If a line is ended with an ampersand (&), the line continues onto the next
line (max. 39 continuation lines allowed)
Semicolon (;) is the separator between statements on a single line

Variables and operators

Variables

Variables must be declared at the beginning of the program or procedure
where they are used

They can also be given a value at declaration (not recommended)
Constants are defined with the PARAMETER attribute – they cannot be
altered after their declaration

Value must be given at declaration
The intrinsic data types in Fortran are integer, real, complex, character
and logical

Numbers and logical Strings

Variables

integer :: n

real :: a, b  ! Multiple variables of same t

complex :: imag_number

logical :: test0, test1

real, parameter :: pi = 3.14159

n = 42

a = 2.2

imag_number = (0.1, 1.0)

test0 = .true.

test1 = .false.

! Automatic conversion between numeric types

b = n + a

! floor operation in real to integer convers

n = b + 3

character :: my_char  ! string with length o

character(len=80) :: my_str

character(len=*), parameter :: name = 'james

! Only first character stored

my_char = 'too long!'

! ' ' and " " are equivalent delimiters

my_str = 'simple string'

! or

my_str = "simple string" 

! including ' and " characters inside a stri

my_str = "this isn't so simple"

my_str = 'is this "complex" string?'

Operators

real :: x, y; integer :: i = 10; logical :: a, b

! Arithmetic operators

x = 2.0**(-i)  ! power function and negation precedence: first

x = x*real(i)  ! multiplication and type change precedence: second

x = x / 2.0    ! division precedence: second

i = i + 1      ! addition precedence: third

i = i - 1      ! subtraction precedence: third

! Relational operators

a < b   ! or (f77) a.lt.b -- less than

a <= b  ! or (f77) a.le.b -- less than or equal to

a == b  ! or (f77) a.eq.b -- equal to

a /= b  ! or (f77) a.ne.b -- not equal to

a > b   ! or (f77) a.gt.b -- greater than

a >= b  ! or (f77) a.ge.b -- greater than or equal to

! Logical operators

.not.b ! logical negation precedence: first

a.and.b ! logical conjunction precedence: second

a.or.b  ! logical inclusive disjunction precedence: third

Arrays

By default, Fortran indexing starts from 1
Programmer can define arbitrary (integer) starting and ending indices

! Arrays

integer, parameter :: m = 100, n = 500

integer :: idx(m)

real :: vector(-n:n)

real :: matrix(m, n)

character(len=80) :: screen(24)

! or

integer, dimension(m) :: idx

real, dimension(-n:n) :: vector

real, dimension(m, n) :: matrix

character(len=80), dimension(24) :: screen

matrix(3, 4) = idx(3) * 2.0

vector(-3) = 4.2

Control structures

Conditionals (if-else)

Conditionals allow the program to execute different code based on some
condition(s)
Condition can be anything from a simple comparison to a complex
combination using logical operators

if (condition) then

  ! do something

else if (condition2) then

  ! .. or maybe alternative something else

else

  ! .. or at least this

end if

Conditionals example

program placetest

  implicit none

  logical :: in_square1, in_square2

  real :: x, y

  write(*,*) ’give point coordinates x and y’

  read (*,*) x, y

  in_square1 = (x >= 0. .and. x <= 2. .and. y >= 0. .and. y <= 2.)

  in_square2 = (x >= 1. .and. x <= 3. .and. y >= 1. .and. y <= 3.)

  if (in_square1 .and. in_square2) then ! inside both

    write(*,*) ’point within both squares’

  else if (in_square1) then ! inside square 1 only

    write(*,*) ’point inside square 1’

  else if (in_square2) then ! inside square 2 only

    write(*,*) ’point inside square 2’

  else ! both are .false.

    write(*,*) ’point outside both squares’

  end if

end program placetest

select case statement allows a
variable to be tested for equality
against a list of values

Only one match is allowed
Usually arguments are character
strings or integers
each case block can contain
multiple statements
default branch if no match found

Select case

integer :: i; logical :: is_prime, & 

                         test_prime_number

! ...

select case (i)

  case (2,3,5,7)

    is_prime = .true.

  case (1,4,6,8:10)

    is_prime = .false.

  case default

    is_prime = test_prime_number(i)

end select

Loops

Three loop formats available in Fortran
integer counter (fixed number of iterations)
condition controlled (do until condition is false)
explicit exit statement

do {control clause}

  ! execute something again and again until stopped

end do

! where the control clause (optional) is either of the form

! i = init_value, max_value, increment

! or a condition to execute while true

! while (condition)

Loops example

 integer :: i, stepsize, numberofpoints 

 integer, parameter :: max_points = 100000

 real :: x_coodinate(max_points), x, totalsum

 

 ! a do-loop with an integer counter (count controlled)

 stepsize = 2

 do i = 1, max_points, stepsize 

   x_coordinate(i) = i*stepsize*0.05 

 end do

 ! condition controlled loop (do while) totalsum = 0.0

 read(*,*) x

 do while (x > 0) 

   totalsum = totalsum + x 

   read(*,*) x 

 end do

Loops example 2

 real :: x, totalsum, eps

 totalsum = 0.0

 

 ! do loop without loop control

 do

   read(*,*) x

   if (x < 0) then

     exit ! = exit the loop

   else if (x > upperlimit) then

     cycle ! = do not execute any further statements, but

           ! instead cycle back to the beginning of the loop

   end if

   totalsum = totalsum + x

 end do

All the control structures (if, 
select, do) can be given names
Construct names can make long
or nested if and select blocks
more readable
cycle and exit can refer to name

Possible to e.g. cycle back to outer
loop from inner loop

Construct names

test: if (m > 0) then

    ...

    ...

end if test

!...

outer: do i=1, 100

   do j = 1, 100

      ...

      if (condition) cycle outer

   end do

end do outer

23

/

Construct names: example

Snippet from  codeFLEUR
...

naloop:DO na2 = na + 1,nat2

            DO i = 1,3

               sum_taual(i) = atoms%taual(i,na) + atoms%taual(i,na2)

            END DO

            DO ix = -2,2

              sum_tau_lat(1) = sum_taual(1) + real(ix)

              DO iy = -2,2

                sum_tau_lat(2) = sum_taual(2) + real(iy)

                DO iz = -2,2

                  sum_tau_lat(3) = sum_taual(3) + real(iz)

                  norm = sqrt(dot_product(matmul(sum_tau_lat,aamat),sum_tau_lat))

                  IF (norm.LT.del) THEN

                     atoms%invsat(na) = 1

                     atoms%invsat(na2) = 2

                     sym%invsatnr(na)  = na2

                     sym%invsatnr(na2) = na

                     WRITE (6,FMT=9000) n,na,na2

                     cycle naloop

                  END IF

               END DO

             END DO

           END DO

END DO naloop

Procedures

Procedures

Procedure is block of code that can be called from other code.
Calling code passes data to procedure via arguments
Fortran has two types of procedures: subroutines and functions

Subroutines pass data back via arguments
Functions return a value

Fortran has both user defined and intrinsic procedures
call random_number(x)   ! Subroutine is preceded by "call" statement, 

                        ! input and output is provided via the arguments

z = sin(x)              ! Function returns value(s) based on the input arguments

Summary

Fortran is – despite its long history – a modern programming language
designed especially for scientific computing

Versatile, quite easy to learn, powerful
In our first encounter, we discussed

Variables, data types, operators
Control structures: loops and conditionals
Fortran has two types of procedures: subroutines and functions

Modules and scoping

Outline

Procedures
Fortran modules
Data scoping

Modular programming

By now, we have implemented the whole application in a single program
block
Larger applications should be divided into small, minimally dependent
modules

Aim is to build complex behavior from simple self-contained components
Modules can be tested and debugged separately
Modules enable easier re-use of code

Procedures

Subroutine

Definition:

Use as:

call sub(act_arg1, act_arg2,...)

Function

Definition:

Use as:

res = func(act_arg1, act_arg2,...)

Procedure definitions

Formally, subroutines and functions are defined and used as:

subroutine sub(dum_arg1, dum_arg2, ...)

  [declarations]

  [statements]

end subroutine sub

[type] function func(dum_arg1, dum_arg2, ...

       &  [result(val)]

  [declarations]

  [statements]

end function func

Either

Function name is valid variable
name within the function scope and
it is used to determine the return
value of the function.

Or

Variable to be used as function
return value is give with result .
Variable has to be declared in the
variable declaration section and the
return type is same as variable type.

Two formats for function definitions

There are two ways to define a function

real function real_func(arg1, arg2)

  real :: arg1, arg2

  real_func = arg1 + arg2

end function real_func

function real_func(arg1, arg2) result(val)

  real :: arg1, arg2, val

  val = arg1 + arg2

end function real_func

Procedure arguments

Types of arguments in procedure definition (dummy arguments) need to
be declared

When calling the procedure, the types of actual arguments need to match
the definition.

Fortran passes arguments by reference
Only the memory addresses of the arguments are passed to the called
procedure
Any change to the value of an argument changes the value at the calling
program

Procedures can have side-effects
The intent attribute can be used to specify how argument is used

Declares how formal argument is
intended to be used for
transferring a value

in: the value of the argument is
read-only i.e. cannot be changed
out: the value of the argument
must be provided
inout (the default)

Compiler uses intent for error
checking and optimization
Improves readability of code

Intent attribute

subroutine foo(x, y, z)

  implicit none

  real, intent(in) :: x

  real, intent(inout) :: y

  real, intent(out) :: z

  x = 10 ! compilation error

  y = 10 ! correct

  z = y * x ! correct

end subroutine foo

Should I use subroutine or function?

Main difference is that functions can be used in expressions:

Recommendation as good programming practice:
Use functions for computing value based on input
Use subroutines for performing operation that changes some of the
inputs

r = dist(x1, y1) + dist(x2, y2)

Modules

Fortran modules

A Fortran module can contain procedures, variables and data structure
definitions
Fortran modules enable

Hiding implementation details
Grouping routines and data structures
Defining generic procedures and custom operators

Module definition and usage

Module is defined with the module keyword and used from main program
or other modules with the use keyword
Depending on the complexity of module, one file can contain a single or
multiple module definitions

Only related modules should be grouped into the same file

Module definition Usage

A simple module example

module geometry

  implicit none

  real, parameter :: pi = 3.14

end module geometry

program testprog

  use geometry

  implicit none

  real :: y

  y = sin(1.2 * pi)

end program testprog

module testmod

  use geometry

  implicit none

  real, parameter :: twopi = 2*pi

Function definition in module Usage

Defining procedures in modules (function)

In most cases, procedures should be defined in modules
Procedures are defined after contains keyword

module geometry

  implicit none

  real, parameter :: pi = 3.14

contains

  real function dist(x, y)

    implicit none

    real :: x, y

    dist = sqrt(x**2 + y**2)

  end function dist

end module geometry

program testprog

  use geometry

  implicit none

  real :: d

  d = dist(2.0, 3.4)

end program testprog

Subroutine definition in module Usage

Defining procedures in modules (subroutine)

In most cases, procedures should be defined in modules
Procedures are defined after contains keyword

module geometry

  implicit none

  real, parameter :: pi = 3.14

contains

  subroutine dist(x, y, d)

    implicit none

    real :: x, y, d

    d = sqrt(x**2 + y**2)

  end subroutine dist

end module geometry

program testprog

  use geometry

  implicit none

  real :: d

  call dist(2.0, 3.4, d)

end program testprog

Scoping

Scoping of variables

Scope is the region of the program where a particular variable is defined
and can be used
A variable with the same name can be defined in multiple scopes and
have different value in them
In Fortran, variables are normally only available within the program unit
that defines them

Local variables in procedures

Local variables can be declared in the procedure
Local variables are not visible outside the procedure
By default, local variables do not retain their values through successive
calls of the procedure

subroutine foo(x, y)

  implicit none

  real, intent(in) :: x

  real, intent(out) :: y

  integer :: i ! Local variable

  ...

variable i starts always from 0 and
gets value 1

variable i gets values 1, 2, 3, … in
each successive call

Local variables in procedures

If local variable is given SAVE attribute, its value is retained through
successive calls
Initialization in declaration is done only in the first call to procedure, and
implicit SAVE is applied

subroutine no_save(x)

  ...

  integer :: i

  i = 0

  i = i + 1

subroutine implicit_save(x)

  ...

  integer :: i = 0

  i = i + 1

Variables in modules

Variables declared in the module definition section are global
Can be accessed from any program unit that uses the module.

Modifications in one program unit are seen also elsewhere

Generally, use of global variables is
not recommended

module globals

  integer :: var

end module commons

...

subroutine foo()

  use globals

  var = 47

end subroutine

...

subroutine bar()

  use globals

  write(*,*) var

end subroutine

...

program myprog

...

  call foo()   ! var is modified

  call bar()   ! The new value is written ou

Limiting visibility of module objects

Variables and procedures in modules can be private or public
public visible for all program units using the module (the default)
private will hide the objects from other program units

module visibility

   real :: x,y

   private :: x

   public :: y ! public declaration is not in principle needed but can improve readability

   real, private :: z ! Visibility can be declared together with type

   private :: foo ! foo can be called only inside this module

 contains

   subroutine foo()

      ...

Protected module variables

Module variables can also be protected
Values of protected variables can be accessed is use-associated scope, but
modifications are prohibited

module protect_demo

   implicit none

   integer, protected :: my_value

 contains

   subroutine modify(val)

     integer :: val

     my_value = val  ! This is ok

   end subroutine modify

end module protect_demo

program demo

  use protect_demo

  implicit none

  call modify(42)

  write(*,*) 'my_value:', my_value

  my_value = 43  ! This fails

  ...

Building modules

Building modules

Each source file needs to be compiled separately
$ gfortran -c mymod.f90 

$ gfortran -c myprog.f90

When compiling the module, a .mod is produced for each module defined
in mymod.f90.
When compiling the main program compiler aborts with error if .mod is
not found for each module used
In order to produce executable, the object files of module and main
program need to be linked

$ gfortran -o myexe mymod.o myprog.o

Building modules

Normally make (or similar build system) is utilized when working with
multiple source files

$ make 

gfortran -c mymod.f90 

gfortran -c myprog.f90 

gfortran -o myexe mymod.o myprog.o

By defining dependencies between program units only those units that
are affected by changes need to be compiled

$ emacs myprog.f90 

$ make 

gfortran -c myprog.f90 

gfortran -o myexe mymod.o myprog.o

Intrinsic modules

53

/

Intrinsic modules

Fortran standard defines several intrinsic modules that are part of the
Fortran language

Most commonly used are iso_fortran_env and iso_c_binding
To make sure that the standard defined version is loaded, you can
specify the intrinsic on use clause:

use, intrinsic :: iso_fortran_env

Precision of built-in numeric types

Fortran standard does not specify the precision of numeric types
(integer, real, complex) i.e. how many bits are used for representing the
number

Default is often 32 bits (real has then 7 significant digits)
The numerical precision can be controlled through the kind parameter:

real(kind=dp) :: number

Value of kind is integer which determines the precision
Before Fortran 2003, the precision was typically defined with the help of 
selected_int_kind and selected_real_kind functions
Nowadays one should use the standard precision types defined in the
intrinsic iso_fortran_env module

Precision of built-in numeric types

The iso_fortran_env module contains several standard precision types
real32, real64, real128, int16, int32, …

It is often a good practice to specify precision in a constant module
variable

Precision can be changed with single modification
module precision

   use, intrinsic :: iso_fortran_env, only : real64, int16 

   integer, parameter :: rp=real64, ip=int16

end module

...

use precision

real(kind=rp):: double_precision_number

integer(kind=ip) :: short_integer_number

Note on precision of constants

Unlike in C, Fortran floating point constants are single precision by default
When double precision constant is needed, use underline suffix with
kind, for example:

use, intrinsic :: iso_fortran_env, only : real64

real(kind=real64) :: value32, value64

value32 = 0.1

value64 = 0.1_real64 ! you may see 0.1D0 in old codes

write(*,'(G0)') value32

write(*,'(G0)') value64

! Result:

! 0.10000000149011612

! 0.10000000000000001

Compiler version and options

In order to make results produced by an application more reproducible,
it can be useful to include in the output information on how the
application was compiled
The iso_fortran_env module contains two functions for that: 
compiler_version and compiler_options

$ gfortran -o test test.f90 -O3 -ffast-math 

$ ./test 

 Compiler: GCC version 4.8.5 20150623 (Red Hat 4.8.5-36) 

 Options: -mtune=generic -march=x86-64 -O3 -ffast-math

use, intrinsic :: iso_fortran_env

write(*,*) 'Compiler: ', compiler_version()

write(*,*) 'Options: ', compiler_options()

Other procedure types and interfaces

Internal procedures

Each program unit (program/subroutine/function) may contain internal
procedures

subroutine mySubroutine

   ...

   call myInternalSubroutine

   ...

 contains

   subroutine myInternalSubroutine

     ...

   end subroutine myInternalSubroutine

end subroutine mySubroutine

Internal procedures

Declared at the end of a program unit after the contains statement
Nested contains statements are not allowed

Variable scoping:
Parent unit’s variables and objects are accessible
Parent unit’s variables are overlapped by local variables with the same
name

Can be called only from declaring program unit
Often used for ”small and local, convenience” procedures

External procedures

Procedures that are defined in a separate non-module program unit are
called external procedures

Old Fortran77 subruotines (e.g. BLAS, LAPACK)
Procedures written with different programming languages

Avoid writing any external procedures in you own code
Module procedures provide much better compile time error checking

Interfaces

It is possible to define separate interfaces to external procedures
Requirement for interfacing procedures written in C

Interfaces are defined by an interface block that can be added to
procedure or module variable declaration section of caller
Interface block consists of one or more procedure declarations with
dummy argument types

No actual executable statements or internal variable declarations

Interface example

! This function is declared outside any modu

function adder(first, second) result(summed)

  real, intent(in) :: first, second

  real :: summed

  

  summed = first + second

  

end function adder

program addertest

  implicit none

  real :: value1, value2, summed

  

  interface

     function adder(val1, val2) result(added

       real, intent(in) :: val1, val2

       real :: added

     end function adder

  end interface

  value1 = 1.0

  value2 = 2.0

  summed = adder(value1, value2)

  print *, summed

end program addertest

Summary

Diving program into modules makes it easier to develop and debug
Fortran modules can contain procedures, variables and type definitions
Procedures should in most cases be implemented in modules
By default, Fortran procedures can modify arguments

Intent attributes can be used for avoiding unwanted side effects

Fortran arrays

66

Outline

Array syntax & array sections
Dynamic memory allocation
Array intrinsic functions
Pointers to arrays

Introduction to Fortran arrays

Fortran is a very versatile programming language for handling arrays
Especially multi-dimensional arrays

Arrays refer to a data type (built-in or derived), and have one or more
dimensions specified in the variable declaration

Fortran supports up to 15 dimensions

Recall: declaring arrays

By default, indexing starts from 1

integer, parameter :: m = 100, n = 500

integer :: idx(m)

real :: vector(0:n-1)

real :: matrix(m, n)

character (len = 80) :: screen(24)

! or equivalently

integer, dimension(m) :: idx

real, dimension(0:n-1) :: vector

real, dimension(m, n) :: matrix

character(len=80), dimension(24) :: screen

Array syntax

Arrays can be accessed element-by-element
Normally, bounds are not checked so access outside declared bounds
results in program crash or garbage results

Arrays can be treated also with array syntax
do j = 0, 10

  vector(j) = 0

  idx(j) = j

end do

vector = 0

! or vector(:) = 0

vector(0:10) = [0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10]

! or

idx(0:10) = [ (j, j = 0, 10) ] ! "implied do"

! the [...] syntax is equivalent to (/ ... /)

Array syntax

Array syntax allows for less explicit do loops

integer :: m = 100, n = 200

real :: a(m,n), x(n), y(m)

integer :: i, j

y = 0.0

do j = 1, n

   do i = 1, m

      y(i) = y(i) + a(i,j) * x(j)

   end do

end do

integer :: m = 100, n = 200

real :: a(m,n), x(n), y(m)

integer :: j

y = 0.0

do j = 1, n

   y(:) = y(:) + a(:,j) * x(j)

end do

Array sections

With Fortran array syntax we can access a part of an array in an intuitive
way: array sections (”slices” in some other languages)
When copying array sections, both left and right hand sides has to have
conforming dimensions

Sub_Vector(3:N) = 0

Every_Third(::3) = 1  ! (start:end:stride)

Diag_Block(i–1:i+1, j–2:j+2) = k

real :: a(1000, 1000)

a(2:500, 3:300:3) = 4.0

lhs(1:3, 0:9) = rhs(-2:0, 20:29) ! this is ok

! but here is an error :

lhs(1:2, 0:9) = rhs(-2:0, 20:29)

Dynamic memory allocation

Dynamic memory allocation

Memory allocation is static if the array dimensions have been declared at
compile time
If the sizes of an array depends on the input to program, its memory
should be allocated at runtime

Memory allocation becomes dynamic

Dynamic memory allocation

integer :: alloc_stat

integer, allocatable :: idx(:)       ! The shapes (but not sizes) of the arrays need to be

real, allocatable :: mat(:,:)        ! known upon declaration

...

read (*,*) m, n

allocate(idx(0:m–1)) ! You can specify the start index, by default indexing starts from 1

allocate(mat(m,n), stat=alloc_stat) ! optional error code to check if allocation

                                     ! was succesful

if (alloc_stat /= 0) stop

...

deallocate(mat) ! Destroys the allocation (and the contents of the array)

75

/

Memory allocation with automatic arrays

”Automatic arrays” can be seen in older codes
Not recommended

subroutine calculate(m, n)

  integer :: m, n ! intended dimensions

  integer :: idx(0:m-1) ! an automatic array

  real :: mat(m,n) ! an automatic array

  ! no explicit allocate – but no checks upon failure either

  ...

  call do_something(m, n, idx, mat)

  ...

  ! no explicit deallocate - memory gets reclaimed automatically

end subroutine calculate

Array intrinsic functions

Array intrinsic functions

Built-in functions can apply various operations on a whole array, not just
array elements

As a result either another array or a scalar value is returned
A selection of just some part of the array through masking is possible

Array intrinsic functions examples

size(array[,dim]) returns the number of elements in the array
shape(array) returns an integer vector containing the size of the array
with respect to each of its dimension
count(l_array[,dim]) returns the count of elements which are .true. in
the logical l_array
sum(array[,dim][,mask]) returns the sum of the elements

optional mask argument is logical array with the same shape as array

Array intrinsic functions

any(l_array[,dim]) returns a scalar value of .true. if any value in logical 
l_array is .true.
all(l_array[,dim]) returns a scalar value of .true. if all values in logical 
l_array are .true.
minval/maxval(array[,dim][,mask]) return the minimum/maximum value
in a given array
minloc/maxloc(array[,mask]) return a vector of location(s) where the
minimum/maximum value(s) are found

Array intrinsic functions: example

integer :: j; integer, parameter :: m = 10, n = 20

real :: x(m,n), v(n)

call random_number(x)

print *, size(x), size(v) ! prints m * n, n

print *, shape(x)         ! prints m, n

print *, size(shape(x))   ! prints 2

print *, count(x >= 0, dim=2)  ! the result is a vector

print *, sum(x, mask=(x < 0.5))  ! performs sum of elements smaller than 0.5

v(1:n) = [ (j, j=1,n) ]

print *, any(v > -1 .and. v < 1)

print *, all(x >= 0, dim=1)

print *, minval(v), maxval(v)

print *, minloc(v), maxloc(v)

Array intrinsic functions

reshape(array,shape) returns a reconstructed array with different shape
than in the input array

integer :: m, n

real :: A(m, n), V(m*n)

...

! convert A (m-by-n matrix) into V (vector of length m*n) without loops

v = reshape(a, shape(v))

Array intrinsic functions

Fortran also includes few basic linear algebra routines
dot_product(v,w) returns a dot product of two vectors
matmul(a,b) returns matrix multiply of two matrices
transpose(a) returns transposed of the input matrix

Performance depends on the compiler and environment

Array intrinsic functions

Array control statements forall and where are commonly used in the
context of manipulating arrays
They can provide a masked assignment of values using effective vector
operations

integer :: j, ix(5)

ix(:) = (/ (j, j=1,size(ix)) /)

where (ix == 0) ix = -9999

where (ix < 0)

  ix = -ix

elsewhere

  ix = 0

end where

integer :: j

real :: a(100,100), b(100), c(100)

! fill in diagonal matrix

forall (j=1:100) a(j,j) = b(j)

! fill in lower bi-diagonal matrix

forall (j=2:100) a(j,j-1) = c(j)

Pointers

Pointers to arrays

The pointer attribute enables creation of array (or scalar) aliasing
variables
Pointer variables are usually employed to refer to another array or array
section

The another array needs to be declared with target attribute
A pointer variable can also be a sole variable itself, used together with 
allocate for dynamic memory allocation

This is not a recommended practice

C programmers: a ”pointer” is a completely different construct in Fortran

Pointers to arrays

Pointers can be used in assignment like normal arrays
integer, pointer :: p(:)  ! Number of dimensions needs to be declared

integer, target :: x(1000)

integer, target, allocatable :: y(:)

...

p => x           ! The pointer array can point to a whole array 

p => x(2:300:5)  ! or a part of it, no need to reallocate

print *, p(1)    ! p(1) points to x(2)

p(2) = 0   ! This would change the value of x(7) too

p => y     ! Now p points to another array y (which needs to be allocated)

nullify(p) ! Now p points to nothing

associated function can be used to
check whether a pointer is
associated with a target

Returns .true. if associated with a
target, .false. if not
For an uninitialized pointer
variables, the return value is
undefined

Pointers to arrays

real, pointer :: p_mat(:,:) => null()

real, target :: mat(100,200)

p_mat => mat

if (associated(p_mat)) then

  print *, 'points to something'

end if

nullify(p_mat)

if (.not.associated(p_mat)) then

  print *, 'points to nothing'

end if

Arrays and procedures

When using arrays as procedure arguments, it is recommended to
always use assumed-shape arrays as dummy argument

Assumed-shape arrays can be only used when procedure interfaces are
explicitly known

subroutine demo(array_a, array_b)

  real, dimension(:,:), intent(in) :: array_a

  integer, intent(in) :: array_b(:)

Summary

Arrays make Fortran language a very versatile vehicle for computationally
intensive program development
Using its array syntax, vectors and matrices can be initialized and used in
a very intuitive way
Dynamic memory allocation enables sizing of arrays according to
particular needs
Array intrinsic functions further simplify coding effort and improve code
readability

Input/Output

Outline

Input/output (I/O) formatting
File I/O

File opening and closing
Writing and reading to/from a file
Formatted and unformatted (binary) files
Stream I/O

Internal I/O
Command line input

Introduction

The long history of Fortran is clearly visible in IO routines
Some concepts are unique and different from any other language
Despite the peculiarities Fortran has powerful IO capabilities

Write statement

Simple form of write statement is

Here the items are the variables whose values are to be written and the
two asterisks denote the unit and formatting

Default unit is the terminal (stdout)
Default formatting depends on the compiler

write(*,*) item1, item2, item3, ...

Input/Output formatting

To prettify output and to make it more readable, use format descriptors
in connection with the write statement
Although less often used nowadays, it can also be used with read to
input data at fixed line positions and using predefined field lengths

95

/

Output formatting

Data type Format descriptors Examples

Integer iw, iw.m write(*,'(i5)') j

write(*,'(i5.3)') j

write(*,'(i0)') j

Real (decimal and fw.d write(*,'(f7.4)') r

exponential forms, ew.d write(*,'(e12.3)') r

auto-scaling) gw.d write(*,'(g20.13)') r

Character a, aw write(*,'(a)') c

Logical lw write(*,'(l)') l

w=width of the output, d=number of digits after the decimal point, m=minimum number of characters

Output formatting: miscellaneous

With complex numbers provide format for both real and imaginary
parts:

Line break and whitespace:

It is possible that an edit descriptor will be repeated a specified number
of times

complex :: z 

write (*,'(f4.1,f4.1)') z ! real and imaginary parts

write(*,'(f6.3,/,f6.3)') x, y ! linebreak between x and y

write(*,'(i3,2x,f6.3)') i, x  ! 2 spaces between i and x

write(*,'(5i8)') ivec(1:5)

write(*,'(4(i5,2x,f8.3))') (ivec(j),zvec(j),j=1,4)

The I0 and G0 format descriptors

Dynamic sizing of real and integer valued output
I0 appeared in F03 and G0 was introduced in F08

Output fields are left justified with all the unnecessary leading blanks
(and precision for real valued variables) removed

Output is <i=12345, reals=1.230000 1.234567890000000>

use iso_fortran_env, only : real32, real64

integer :: i = 12345

real (kind=real32) :: sp = 1.23_real32

real (kind=real64) :: dp = 1.234567890_real64

write(*,fmt='("<i=",i0,", reals=",g0,1x,g0,">")') i,sp,dp

Handling character strings

Fortran provides several intrinsic functions for handling character
strings, such as

trim(string) - removes blank spaces from the end of string
adjustl(string)/adjustr(string) - moves blank spaces from the
beginning/end of the string to the end/beginning of it
len(string) - length of a string
index(string, substring) - returns the starting position of a substring
within a string

File I/O

Opening and closing files: basic concepts

By now, we have written to terminal or read from keyboard with 
write(*, ...) and read(*, ...)
Writing to or reading from a file is similar with few differences:

File must be opened with an OPEN statement, in which the unit number
and (optionally) the file name are given
Subsequent writes (or reads) must refer to the given unit number
File should be closed at the end

Opening and closing a file

The syntax is (the brackets [ ] indicate optional keywords or arguments)
open([unit=]iu, file='name' [, options]) 

close([unit=]iu [, options])

For example
open(10, file= 'output.dat', status='new')

close(unit=10, status='keep')

Opening and closing a file

The first parameter is the unit number
The keyword unit= can be omitted

The unit numbers 0, 5 and 6 are predefined
0 is output for standard (system) error messages
5 is for standard (user) input
6 is for standard (user) output
These units are opened by default and should not be re-opened nor
closed by the user

If the file name is omitted in the OPEN, the file name will based on unit
number, e.g. ’fort.12’ for unit=12

File opening options

status: existence of a file
‘old’, ‘new’, ‘replace’, ‘scratch’, ‘unknown’

position: offset, where to start writing
‘append’

action: file operation mode
‘write’, ‘read’, ‘readwrite’

form: text or binary file
‘formatted’, ‘unformatted’

File opening options

access: direct or sequential file access
‘direct’, ‘sequential’, ‘stream’,

iostat: error indicator, (output) integer
non-zero only upon an error

recl: record length, (input) integer
for direct access files only
warning (check): may be in bytes or words

File opening: file properties

Use the inquire statement to find out information about
file existence
file unit open status
various file attributes

The syntax has two forms, one based on file name, the other for unit
number

inquire(file='name', options ...) or
inquire(unit=iu, options ...)

File opening: file properties

exist does file exist? (logical)

opened is file / unit opened? (logical)

form ‘formatted’ or ‘unformatted’ (char)

access ‘sequential’ or ‘direct’ or ‘stream’ (char)

action ‘read’, ‘write’, ‘readwrite’ (char)

recl record length (integer)

size file size in bytes (integer)

File opening: file properties example

! check the existence of a file

logical :: file_exist

inquire(file='foo.dat', exist=file_exist)

if (.not. file_exist) then

  write(*,*) 'the file does not exist'

else

  ! do something with the file foo.dat

endif

File writing and reading

Writing to and reading from a file is done by giving the corresponding
unit number (iu) as a parameter :

Formats and other options can be used as needed
If keyword unit is used, also the keyword fmt must be used

Note: fmt is applicable to formatted, text files only
The star format (*) indicates list-directed output (i.e. programmer does
not choose the input/output styles)

write(iu,*) str

write(unit=iu,fmt=*) str

read(iu,*) str

read(unit=iu, fmt=*) str

Formatted vs. unformatted files

Text or formatted files are
Human readable
Portable i.e. machine independent

Binary or unformatted files are
Machine readable only, generally not portable
Much faster to access than formatted files
Suitable for large amount of data due to reduced file sizes
Internal data representation used for numbers, thus no number
conversion, no rounding of errors compared to formatted data

Unformatted I/O

Write to a sequential binary file

No format descriptors allowed
Reading similarly

real :: rval

character(len=60) :: string

open(10, file='foo.dat', form='unformatted')

write(10) rval

write(10) string

close(10)

read(10) rval

read(10) string

Stream I/O

A binary file write adds extra record delimiters (hidden from
programmer) to the beginning and end of records
In Fortran 2003 using access method ‘stream’ avoids this and
implements a C-like approach
It is recommended to use stream I/O
Create a stream (binary) file

Reading similarly

real :: dbheader(20), dbdata(300)

open(10,file='my_database.dat', access='stream')

write(10) dbheader

Often it is necessary to filter out
data from a given character string

Or to pack values into a character
string

Fortran internal I/O with read and 
write becomes now handy:
instead of unit number one
provides character array as first
argument

Actual files are not involved at all

Internal I/O

character(len=13) :: s1

character(len=60) :: s2

integer :: njobs, istep

! extract a number from character string 

s1 = 'time step# 10'

read(s1,fmt='(10x,i3)') istep

! write data to a character string

njobs = 2014

write(s2,'(a,i0)') '# of jobs completed = ',

     & njobs

Command line input

Command line input

In many cases, it is convenient to give parameters for the program
directly during program launch

Instead of using a parser, reading from an input file etc.
Fortran 2003 provides a way for this

command_argument_count()

get_command_argument()

Example: reading in two integer
values from the command line
The (full) program should be
launched as (e.g.):

$ ./a.out 100 100

Command line input

subroutine read_command_line(height, width)

  integer, intent(out) :: height, width

  character(len=10) :: args(2)

  integer :: n_args, i

  n_args = command_argument_count()

  if ( n_args /= 2 ) then

    write(*,*) ' Usage : ./exe height width'

    call abort()

  end if

  do i = 1, 2

    call get_command_argument(i, args(i))

    args(i) = trim(adjustl(args(i)))

  end do

  read(args(1),*) height

  read(args(2),*) width

end subroutine read_command_line

Summary

Input/Output formatting
Files: communication between a program and the outside world

Opening and closing a file
Data reading & writing

Use unformatted (stream) I/O for all except text files

User defined data types

Outline

Defining own datatypes
Scope of definitions
Using datatypes

Nested types
Memory layout

What is a derived data type?

Derived data type is a data structure composed of built-in data types and
possibly other derived data types

Equivalent to structs in C programming language
Derived type is defined in the variable declaration section of program
unit

Not visible to other programming units
Normally, should be defined in a module and used via the use clause

What is a derived data type?

For real-world applications, using only intrinsic types is often insufficient
It is beneficial to group the data together as larger objects

Code becomes easier to read and maintain
Fewer arguments to procedures
Encapsulation of data

Variables used in the same context should be grouped together using
modules and derived data types

121



/

Derived type declaration

Type declaration

Declaring a variable with a type

Elements are accessed with % operator

type particletype

  real :: x 

  real :: y

  real :: z

  integer :: charge 

end type particletype

type(particletype) :: proton, sodiums(300)

write(*,*) proton % charge

Visibility of derived datatypes (scoping)

When declared in the same programming unit, derived data types are
visible to that unit only
When declared in a module unit, a derived data type can be accessed
and used outside the module through use statement

Derived type initialization

Data type initialization
sodiums(1) = particletype(0.75, 0.19, 0.0, 1)

sodiums(2) = particletype(0.0, -0.38, 0.0, 8) 

sodiums(3) = particletype(-0.75, 0.19, 0.0, 1) 

! or elementwise

sodiums(1) % x = 0.75 

sodiums(1) % y = 0.19 

sodiums(1) % z = 0.0

Nested derived types

Derived types can contain other derived types as components

Access as

type moleculetype 

  type(particletype), allocatable :: atoms(:) 

  real :: mu

end type moleculetype 

!... 

type solvent 

  type(moleculetype), allocatable :: fluid(:)

  complex :: epsilon 

end type solvent

beverage % fluid(1) % atoms(1) % x = 0.75

Array of Structures Structure of Arrays

Data structures: memory layout

type point 

  real :: x, y, z

end type point

type(point), allocatable :: points(:)

allocate(points(N))

type point 

  real, allocatable :: x(:) 

  real, allocatable :: y(:) 

  real, allocatable :: z(:)

end type point

type(point) :: points

allocate(points%x(N), & 

         points%y(N), & 

         points%z(N))

Array of Structures Structure of Arrays

Data structures: memory layout

integer :: i, j

real :: dist(4,4)

do i = 1, 4

  do j = i, 4 

    dist(i,j) = sqrt( & 

      (points(i)%x-points(j)%x)**2 + &

      (points(i)%y-points(j)%y)**2 + &

      (points(i)%z-points(j)%z)**2)

  end do

end do

integer :: i, j

real :: dist(4,4)

do i = 1, 4

  do j = i, 4 

    dist(i,j) = sqrt( & 

      (points%x(i)-points%x(j))**2 + &

      (points%y(i)-points%y(j))**2 + &

      (points%z(i)-points%z(j))**2)

  end do

end do

Summary

Derived data types enables grouping of data to form logical objects
A Fortran program becomes more readable and modular with sensible
use of derived data types
Handling of complex data structures such as linked lists or binary trees
becomes more manageable with use of derived types
Enables the use of object oriented programming concepts

Some further useful features in Fortran

Outline

Generic procedures
Procedure pointers
Environment variables
Executing commands

Generic procedures

Procedures which perform similar actions but for different data types
can be defined as generic procedures
Procedures are called using the generic name and compiler uses the
correct procedure based on the argument number, type and dimensions

Generic name is defined in an interface section
Can improve code readability and maintainability

Overuse can have the opposite effect

Generic procedures example

module swapmod

   implicit none

   interface swap

      module procedure swap_real, swap_char

   end interface swap

 contains

   subroutine swap_real(a, b)

     real, intent(inout) :: a, b

     real :: temp

     temp = a; a = b; b = temp

   end subroutine swap_real

   subroutine swap_char(a, b)

     character, intent(inout) :: a, b

     character :: temp

     temp = a; a = b; b = temp

   end subroutine swap_char

end module swapmod

program switch

  use swapmod

  implicit none

  character :: n,s

  real :: x,y

  n = 'j'

  s = 's'

  x = 10

  y = 20

  print *, x, y, n, s

  call swap(n, s)

  call swap(x, y)

  print *, x, y, n, s

end program switch

Overloading operators

Procedures can also be assigned to operators
For example *, +, -, ==, <, >
Own operator names can be used too, for example .dot.

Especially useful together with user-defined data types

Overloading operators

Instead of a procedure name, use operator or assignment keyword
Operator: 1 or 2 input arguments with intent(in)
Assignment: 2 input arguments with intent(inout) or intent(out) and
intent(in)

interface operator(+)

  module procedure base_sum

end interface operator(+)

function base_sum(x,y) result(z)

  type(base), intent(in) :: x,y

  type(base) :: z

  z % val = x % val + y % val

end function

Operator overloading example

module operator_demo

   implicit none

   type pair

      real :: a, b

   end type pair

   interface operator (<)

      module procedure is_smaller

   end interface operator (<)

 contains

   logical function is_smaller(x, y)

     type(pair), intent(inout) :: x, y

     if (x%a < y%a .and. x%b < y%b) then

        is_smaller = .true.

     else

        is_smaller = .false.

     end if

   end subroutine

end module operator_demo

program switch

  use operator_demo

  implicit none

  type(pair) :: p1, p2

  p1%a = 1.0; p1%b = 2.0

  p2%a = 3.0; p2%b = 4.0

  if (p1 < p2) then

     print *, ‘p1 is smaller’

  else

     print *, ‘p1 not smaller’

  end if

end program switch

Procedure pointers

It is possible to pass procedures
as arguments

Somewhat similar to C function
pointer

Here add_one is passed as
argument

Interface of the procedure is
known because implementation
of function is known to compiler

Procedures as arguments

module test

 contains

  function apply(f, x) result(y)

    procedure(add_one) :: f

    integer, intent(in) :: x

    integer :: y

    y = f(x)

  end function apply

  function add_one(x) result(y)

    integer, intent(in) :: x

    integer :: y

    y = x + 1

  end function add_one

end module test

program test

  use test

  print *, apply(add_one, 10)

end program test

Instead of explicit module
function one can also provide an 
interface as type for f in apply
function
Fortran has also abstract
interfaces

Routines defined in abstract
interface must now have an
actual implementation in the
current scope

Procedure interfaces

module test

  interface

     fuction int_fun(x) result(y)

     integer, intent(in) :: x

     integer :: y

  end interface

contains

  function apply(f, x) result(y)

    procedure(int_fun) :: f

    integer, intent(in) :: x

    integer :: y

    y = f(x)

  end function apply

end module test

138

/

With procedure statement one can
define also pointers to procedures
(explicit or implicit interfaces)
A user-defined type can also have
procedure pointers as
components

Procedure pointers

procedure(int_fun), pointer :: fptr

fptr => add_one

print *, fptr(1)

fptr => substract_one

print *, fptr(1)

function substract_one(x) result(y)

  integer, intent(in) :: x

  integer :: y

  y = x - 1

end function substract_one

function add_one(x) result(y)

  integer, intent(in) :: x

  integer :: y

  y = x + 1

end function add_one

Environment variables

140



/

Environment variables

Besides command line arguments, environment variables are a common
way to modify program behavior
Fortran has a standardized method for accessing values of environment
variables

Environment variables

Besides command line arguments, environment variables are a common
way to modify program behavior

character(len=256) :: enval

integer:: length, stat

! extract hostname

call get_environment_variable('HOSTNAME', enval, length, stat)

if (stat == 0) write (*,'(a,a)') 'host=', enval(1:len)

! extract user

call get_environment_variable('USER', enval, length, stat)

if (stat == 0) write (*,'(a,a)') 'user=', enval(1:len)

Executing commands

Executing commands

Invoking external programs from within a program may be useful
occasionally

No source nor library API available for a useful program
Perl/Python/etc parsing scripts

Fortran (2008) has a standardized method for invoking an external
command

Executing commands

call execute_command_line(command[,wait][,exitstat][,cmdstat]

[,cmdmsg])

command is a character string containing the command to be invoked
wait is logical value indicating if command termination is to be waited
(.true., the default) or if the command is to be executed asynchronously
(.false.) (optional)
exitstat is an integer value containing the return value of the command if
wait=.true. (optional)
cmdstat is an integer value. It is assigned a value of zero if command
executed successfully. For other return codes, see docs (optional)
cmdmsg is a character string containing explanatory message for positive
values of cmdstat (optional)

Executing commands: example

program execcommand

  implicit none

  integer :: estat, cstat

  ! execute a unix command

  call execute_command_line('ls -al', .true., estat, cstat)

  if (estat==0) write (*,'(a)') 'command completed successfully’

end program execcommand

Summary

Generic procedures
Same procedure name for multiple argument types
Can improve code readability

Operator overloading
Can use operators for user-defined types

Procedure pointers
Reading environment variables
Executing system commands

Advanced features in Fortran

Outline

Language interoperability
Object oriented programming
Fortran coarrays

Language interoperability issues

The problem of language interoperability has been present for a long
time

Before Fortran 2003 the standard did not specify the calling convention in
detail

Interoperability is becoming even more important than before
Steering of computation with scripting languages
Utilizing libraries that are written using other programming languages

Before Fortran 2003

Compilers often, but not always
Passed all procedure arguments by-reference (i.e. by-address)
Referred to functions and subroutines in lowercase letters and by adding
an additional underscore after, e.g. Func becomes func_ to be found by
the linker
Passed function return values via stack
Passed character strings by-reference, with an additional hidden length
argument, passed by value.
Note: Fortran character strings are not null-character terminated

Portability

The traditional way to have interoperability with C requires a priori
knowledge of lowercase and underscore policy used by the compiler
Complex cases, such as passing character strings or passing arguments
by value, are generally very error prone and may lead to catastrophic
errors at runtime
Often a separate layer of C functions was needed for interoperability

The iso_c_binding module

Fortran 2003 provides intrinsic module iso_c_binding for interoperation
with C.

Interoperability with other languages (e.g. Python) can be often done via
C

Module contains
Access to named constants that represent kind type parameters of data
representations compatible with C types
The derived types c_ptr and c_funptr corresponding to C pointer and C
function pointer types, respectively
Useful procedures: c_loc, c_funloc, c_f_pointer, c_associated, 
c_f_funpointer, c_sizeof (f08)

Interoperability with Fortran 2003: example

function call_c_func(data, nx, ny) result(stat)

  use, intrinsic :: ISO_C_BINDING

  implicit none

  real(dp), dimension(:,:), intent(in) :: data

  integer, intent(in) :: nx, ny

  integer :: stat

  interface

     ! C-definition is: int func(double *data, const int nx, const int ny)

     function c_func(data, nx, ny) bind(C,name="func") result(stat)

       use, intrinsic :: ISO_C_BINDING

       implicit none

       real(kind=C_DOUBLE) :: data(*)

       integer(kind=C_INT), value, intent(IN) :: nx, ny

       integer(kind=C_INT) :: stat

     end function c_func

  end interface

  stat = c_func(data, nx, ny)

end function call_c_func

Calling C routines

A Fortran subroutine maps to a C function with void result
A Fortran function maps to a C function returning a value
Binding label in bind(c, name=<label>)

The routine is known to the C compiler as specified by the binding label
By default the Fortran name in lower case

Mapping of C intrinsic data types

Interoperable mappings for the most commonly used C intrinsic data
types (note that Fortran does not support unsigned integers)

Fortran declaration C data type

integer(c_short) short int

integer(c_int) int

integer(c_long) long int

real(c_float) float

real(c_double) double

character(1,c_char) char

Value attribute

For scalar dummy arguments, the value attribute can be used to pass the
value to the procedure

Actual argument is copied
Dummy argument can be altered but result is not visible to caller
Argument must not be pointer, allocatable, have intent(out) or 
intent(inout), be a procedure

value attribute is not limited to procedures with bind attribute

Other interoperability features

Concept of a pointer is completely different in Fortran and C
In Fortran, pointer is a storage descriptor that holds the address of a
target and in case of arrays, information on dimension and bounds
C pointer is an memory address with associated type

Module iso_c_bindings contains Fortran type definitions for C pointers
and function pointers

type(c_ptr) abstracts C pointers and
type(c_fun_ptr) abstract C function pointers

These are needed when manipulating raw C pointers

Object oriented programming

Object Oriented Programming (OOP) is programming paradigm where
Data and functionality are wrapped inside of an “object”
Objects provide methods which operate on (the data of) the object
Method is a procedure that is tied to the data of an object
Objects can be inherited from other “parent” objects

Object oriented programming in Fortran is facilitated with type
extensions and type- and object-bound procedures

Fortran coarrays

Parallel processing as part of Fortran language standard
Only small changes required to convert existing Fortran code to support a
robust and potentially efficient parallelism

A Partitioned Global Address Space (PGAS) paradigm
Parallelism implemented over “distributed shared memory”

Integrated into Fortran 2008 standard
Compiler support and especially performance varies

Summary

Fortran is a modern language that continues to evolve
Support for object-oriented programming
Complex, generic data structures can be implemented in modern Fortran
Coarrays provide parallel programming support in the language itself

Fortran remains as one of the major languages for scientific computation

